Maple-assisted proof of formula for A251367

Robert Israel

19 March 2018

Of the $3^6 = 729$ possible configurations for a 2×3 subblock (where $[a, b, c, d, e, f]$ encodes $\begin{bmatrix} a & b & c \\ d & e & f \end{bmatrix}$), 695 have both 2 x 2 subblocks summing to 1 to 7.

```
> Blockconfigs:= select(t -> {t[1]+t[2]+t[4]+t[5], t[2]+t[3]+t[5]+t
  [6]} subset {$1..7}, [seq(convert(x,base,3)[1..6],x=3^6+1.
  .3^6+3^6)]):
  nops(Blockconfigs);
```

$$695 \tag{1}$$

Consider the 695×695 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×3 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j), and 0 otherwise. The following Maple code computes it.

```
> T:= Matrix(695,695,(i,j) -> `if`(Blockconfigs[i][1..3]=
  Blockconfigs[j][4..6],1,0));
```

$$T := \begin{bmatrix} \text{695 x 695 Matrix} \\ \text{Data Type: anything} \\ \text{Storage: rectangular} \\ \text{Order: Fortran\_order} \end{bmatrix} \tag{2}$$

Thus for $n \geq 1$, $a(n) = u\, T^{n-1}\, v$ where u and v are 695-dimensional row and column vectors respectively of all 1's. The following Maple code produces these vectors.

```
> u:= Vector[row](695, 1):
  v:= Vector(695,1):
```

To check, here are the first few entries of our sequence.

```
> vn[0]:= v:
  for n from 1 to 11 do vn[n]:= T . vn[n-1] od:
  seq(u . vn[n-1],n=1..11);
```

$$695, 17969, 464393, 12002283, 310199103, 8017100977, 207202101873, 5355141623323, 138403720518311, 3577046360518609, 92448820142650873 \tag{3}$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= LinearAlgebra:-MinimalPolynomial(T, t);
```

$$P := t^8 - 19\, t^7 - 165\, t^6 - 311\, t^5 + 70\, t^4 + 332\, t^3 - 136\, t^2 \tag{4}$$

The empirical formula corresponds to a factor of this: $t^3 - 24\, t^2 - 49\, t + 34$.

```
> Q:= normal(P/(t^3-24*t^2-49*t+34));
```

$$Q := \left(t^3 + 5\, t^2 + 4\, t - 4\right) t^2 \tag{5}$$

If

$r(n) = a(n+3) - 24\, a(n+2) - 49\, a(n+1) + 34\, a(n) = u \cdot \left(T^{n+2} - 24\, T^{n+1} - 49\, T^n\right.$

$+ 34\, T^{n-1})\, v$

, then $r(n+3) + 5\, r(n+2) + 4\, r(n+1) - 4\, r(n) = u\, T^{n-3}\, P(t)\, v = 0$ for $n \geq 3$.

In order to check that the empirical formula is true, it suffices to verify $r(n) = 0$ for $n = 1 \ldots 5$.

```
> seq(u . (vn[n+2] - 24*vn[n+1] - 49*vn[n] + 34*vn[n-1]), n=1..5);
```

$$0, 0, 0, 0, 0 \tag{6}$$